
SQLCA —Communication Area O

®» ['|3 struttura dati fondamentale con cui SQL ci restituisce
informazioni sull'esito di qualsiasi istruzione SQL

® |n RPG e definita dal precompilatore SQL
= Viene valorizzata dopo ogni istruzione SQL

®» E'da utilizzare obbligatoriamente per conoscere l'esito
dell'istruzione (tranne istruzioni DECLARE). Embedded SQL non

restituira mai un'eccezione RPG come con I/O nativo

= |n ogni programma esiste una sola SQLCA

SQL Diagnostics Area i

® F'uninsieme di informazioni aggiuntive rispetto a quanto
restituito in SQLCA

® \engono reperite tramite l'istruzione SOL get diagnostics

SQLDA - 5SQL Descriptor Area

® ' uninsieme di variabili utilizzate nell'istruzione describe
oppure dalle istruzioni prepare, open, call, fetche
execute

® |n RPG I'SQLDA e definita dal precompilatore SQL

® |'SQLDA utilizzata in describe o prepare restituisce
informazioni sull'istruzione select che e stata preparata

SQLDA - definizione

® Sj definisce con l'istruzione SQL include sqglda

® crea le strutture dati SQLDA, SQLVAR, SQLVAR2

® || sottocampo SQL VAR di SQLDA € un array con numero
elementi SQL NUM

™ SQL NUM va definito come una costante numerica con la
dimensione richiesta per I'array SQL. VAR

SQLDA - header

» Nella parte di intestazione dell'SQLDA ci sono 4 variabili

campo tipo describe/prepare fetch, open, call, execute

(impostato dal DB2 (impostato da utente prima
tranne SQLN) dell'esecuzione)

SQLDAID char(8) 'SQLDA' 7°byte: "2, '3, 4"
numero di occorrenze di SQLVAR
per ogni colonna del resultset

SQLDABC int lunghezza SQLDA da settare con valore >=
16+SQLN*80
SQLN smallint da settare con valore >= occorrenze di SQLVAR
0 prima dell'esecuzione
SQLD smallint numero colonne occorrenze di SQLVAR
descritte da SQLVAR

SOLDA - SQLVAR occurences

= Ogni occorrenza contiene 5 variabili
= | unghezza totale di ogni occorrenza: 8o bytes

® Per |e istruzioni prepare e describe ogni occorrenza descrive

una colonna del resultset

Manipolare il descrittore

» A partire dallaVgR4 il DB2 agevola l'utilizzo della
struttura SQLDA tramite l'uso dell'SQL descriptor

® allocate descriptor,deallocate descriptor

® get descriptor, set descriptor reperisce oimposta le
informazioni di un descrittore

® using sqgl descriptor clausola daaggiungere su istruzione open
per utilizzare il descrittore precedentemente impostato

® into sgl descriptor clausola daaggiungere su istruzione fetch
per indirizzare I'ouput al descrittore precedentemente impostato

® describe istruzione per reperire le informazioni di definizione di una
tabella o del resultset di un'istruzione sql di select

® queste istruzioni possono essere utilizzate solo in embedded SQL

Descrittore e SQL dinamico/1

Codice da:
Ufficio: D11
(blank = tutti)

Codice da:
Ufficio:
(blank = tutti)

Codice da: 000010
Ufficio:

Codice da: 000010
Ufficio: D11
(blank = tutti)

(blank = tutti)

XXX XX XXXXXXX
XXX XX XXXXXXX

XXX XX XXXXXXX
XXX XX XXXXXXX

XXX XX XXXXXXX
XXX XX XXXXXXX

XXX XX XXXXXXX
XXX XX XXXXXXX

select * select * select * select *
from EMPLOYEE from EMPLOYEE from EMPLOYEE from EMPLOYEE
order by EMPNO; where EMPNO >= :WSEMPNO where WORKDEPT = :WSDEPT where EMPNO >= :WSEMPNO
order by EMPNO; order by EMPNO; and WORKDEPT = :WSDEPT

order by EMPNO;
select *

SO IO unica istruzione ma con
where EMPNO >= :WSEMPNO

and WORKDEPT between :WKDEPT1 and :WKDEPT2 ‘ probabili decadenza di
order by EMPNO; performance




Descrittore e SQL dinamico/2

® | a soluzione con SQL dinamico avrebbe il vantaggio di ottimizzare
la scrittura dell'istruzione per ottenere le migliori performance

®» comporterebbe una maggiore attenzione nella scrittura del codice
per concatenare correttamente costanti e contenuto delle variabili
dei campi a video (e relativi apici!)

dcl-s sglstm varchar (1000);
dcl-s sqglstmW like(sglstm); ATTENZIONE A:
dcl-c AP const('''"); . .
- concatenazioni
sglstm = 'select * from EMPLOYEE ' apiciedoppi
if WSEMPNO <> *blanks; .
sglstmW = ' where EMPNO >= ' + AP + %trim (WSEMPNO) + AP; aF”CI
endif; - conversione
if WSDEPT <> *blanks; . oMo
if sglstmW = *blanks; Varlablll
sqlstmW = ' where'; numeriche e
endif; .
sglstmW = sglstmW + ' and WORKDEPT = ' + AP + %trim (WSDEPT) + AP; data/ora In
endif; alfanumeriche
sglstm = sglstm + ' order by EMPNO';




Descrittore e SQL dinamico/3

® F'sicuramente preferibile usare SQL dinamico con i parameter
markers (?).
® quanti sono i parameter markers da sostituire con using all'apertura del
cursore?

® 3 quali tipi di variabili corrispondono i parameter marker?
nel
medesimo

® cjajuta l'utilizzo del descrittore:

® 3llocazione descrittore

ordine
posizionale

® aggiunta condizione where ove richiesto con parameter marker
® aggiunta del valore di confronto di ogni parametro al descrittore
™ apertura cursore con il descrittore

select * select * select * select *

from EMPLOYEE
order by EMPNO;

from EMPLOYEE

where EMPNO >= %
order by EMPNO;

from EMPLOYEE
where WORKDEPT
order by EMPNO;

?

from EMPLOYEE

where EMPNO >= ?

and WORKDEPT
order by EMPNO;




Descrittore e SQL dinamico/y4

numero max

= allocazione del descrittore dei parameter
exec sql // markers
_ allocate descriptor local 'dCursl' with max 2; del
local: ambito S == NOME A€
interno al ® preparazione per costruzione istruzioneSQL\, descrittore
programma. dcl-s sglstm varchar (1000); AT_-I-!ENZIONE agli
global:amblto dcl-s sqlstmil like (sqlstm) ; apici e al case-
della sessione SQL dcl-s NumParm uns (3) ; sensitive!
quindi puo essere sqlstm = 'select EMPNO, LASTNAME, FIRSTNME + E' meglio usare una
usato anche in un from EMPLOYEE '; variabile host

altro programma

® aggiunta filtri di selezione

| type
i1f WSEMPNO <> *blanks; 1 = character
sglstmW = ' where EMPNO >= ?'; _ .
NumParm += 1; 2= nurT\erlc
exec sgl 3=deC|maI
: : set descriptor 'dCursl' = date/time
Esempla”egatl: value :NumParm type = 1, length = 6, data = :WSEMPNO; > _ /h
DESCO1.SQLRPGLE, | ... - -z = VEllerEl
DESCO1B.SQLRPGLE

Descrittore e SQL dinamico/s

® imposto il numero di parameter marker effettivamente inseriti

nell'istruzione SQL Pt  ATTENZIONE!
exec sal / Se si usa una costante
4 come nome descrittore

non corrispondenti nelle
® preparo l'istruzione SQL (prepare varie istruzioni si riceve
p p (p p ) errore SQLOQO77
. ‘- . . "N
= apro il cursore utilizzando il descrittore eeeritiore non

corretto"
exec sqgl

open Cursl using SQL descriptor 'dCursl';

®» |eggo il cursore (fetch) e chiudoiil cursore (c1lose) e poi disalloco

il descrittore. Il descrittore viene implicitamente disallocato insieme alla
chiusura del cursore.

exec sqgl
deallocate descriptor local 'dCursl';

Un passo avanti...

® Questo e uno dei possibili modi per utilizzare il descrittore SQL.

® Nei prossimi Power Coffee vedremo altri esempi

// ESEMPIO UTILIZZO SQL DESCRIPTOR CON SQL DINAMICO CON PARAMETER MARKERS

// (c) MarkOne 2022, www.markonetools.it



// esempio di costruzione di una istruzione SQL dinamica in base a due

//  parametri (codice impiegato e codice ufficio)

// per semplificare l'esempio sono state omesse tutte le parti di controllo

//  dell'esito delle istruzioni SQL



// Curs1: cursore sql

// dCurs1: descrittore sql

// pCurs1: istruzione preparata per il cursore Curs1 in base alla stringa sqlstm



//--- MAIN INTERFACE ---

dcl-pr DESC01 extpgm('DESC01');

 *n char(6);

 *n char(3);

end-pr;

dcl-pi DESC01;

 WSEMPNO char(6);

 WSDEPT char(3);

end-pi;



dcl-ds EMPLOYEE ext template end-ds;



// istruzione SQL dinamica

dcl-s sqlstm varchar(1000);

  // porzione where dell'istruzione SQL

dcl-s sqlstmW like(sqlstm);

// numero di parameter markers effettivamente utilizzati

dcl-s NumParm uns(3) inz(*zeros);



// output istruzione SQL

dcl-ds dsOut qualified;

  Codice like(EMPNO);

  Cognome like(LASTNAME);

  Nome like(FIRSTNME);

end-ds;

dcl-s Msg char(52);



// dichiarazione cursore

exec sql

  declare Curs1 cursor for pCurs1;



// allocazione descrittore SQL con massimo 2 elementi

//  ovverosia il numero massimo di parameter marker

exec sql

  allocate descriptor local 'dcurs1' with max 2;



// costruzione istruzione SQL e impostazione elementi del descrittore

sqlstm = 'select EMPNO, LASTNAME, FIRSTNME from EMPLOYEE';

  // filtro codice impiegato

if WSEMPNO <> *blanks;

  if sqlstmW = *blanks;

    sqlstmW = ' where EMPNO >= ?';

  else;

    sqlstmW = sqlstmW + ' and EMPNO >= ?';

  endif;

  NumParm += 1;

   exec sql

    set descriptor 'dCurs1'

       value :NumParm type = 1, length = 6, data = :WSEMPNO;

endif;

  // filtro ufficio

if WSDEPT <> *blanks;

  if sqlstmW = *blanks;

    sqlstmW = ' where WORKDEPT = ?';

  else;

    sqlstmW = sqlstmW + ' and WORKDEPT = ?';

  endif;

  NumParm += 1;

   exec sql

    set descriptor 'dCurs1'

       value :NumParm type = 1, length = 3, data = :WSDEPT;

endif;

  // istruzione sql finale

sqlstm = sqlstm + sqlstmW + ' order by EMPNO';



  // numero totale di parameter marker ovverosia numero totale

  //  di elementi aggiunti al descrittore

exec sql

  set descriptor 'dCurs1' count = :NumParm;



// preparazione

exec sql

  prepare pCurs1 from :sqlstm;



// apertura cursore utilizzando il descrittore SQL

//  per valorizzare i parameter markers

exec sql

  open Curs1 using sql descriptor 'dCurs1';



dou sqlcode = 100;

  exec sql

    fetch next from Curs1 into :dsOut;



  if sqlcode = 100;

    leave;

  endif;



  Msg = dsOut.Codice + ' ' + %trim(dsOut.Cognome) + ' ' + %trim(dsOut.Nome);

  dsply Msg;



enddo;



exec sql

  close Curs1;



exec sql

  deallocate descriptor local 'dCurs1';



*inlr = *on;


**free



// ESEMPIO UTILIZZO SQL DESCRIPTOR CON SQL DINAMICO CON PARAMETER MARKERS

// (c) MarkOne 2022, www.markonetools.it



// esempio di costruzione di una istruzione SQL dinamica in base a due

//  parametri (codice impiegato e codice ufficio)

// per semplificare l'esempio sono state omesse tutte le parti di controllo

//  dell'esito delle istruzioni SQL



// Curs1: cursore sql

// dCurs1: descrittore sql (referenziato con variabile NomeDesc)

// pCurs1: istruzione preparata per il cursore Curs1 in base alla stringa sqlstm



//--- MAIN INTERFACE ---

dcl-pr DESC01B extpgm('DESC01B');

 *n char(6);

 *n char(3);

end-pr;

dcl-pi DESC01B;

 WSEMPNO char(6);

 WSDEPT char(3);

end-pi;



dcl-ds EMPLOYEE ext template end-ds;



// istruzione SQL dinamica

dcl-s sqlstm varchar(1000);

  // porzione where dell'istruzione SQL

dcl-s sqlstmW like(sqlstm);

// numero di parameter markers effettivamente utilizzati

dcl-s NumParm uns(3) inz(*zeros);



// output istruzione SQL

dcl-ds dsOut qualified;

  Codice like(EMPNO);

  Cognome like(LASTNAME);

  Nome like(FIRSTNME);

end-ds;

dcl-s Msg char(52);



// nome descrittore

dcl-s NomeDesc varchar(20) inz('dCurs1');



// dichiarazione cursore

exec sql

  declare Curs1 cursor for pCurs1;



// allocazione descrittore SQL con massimo 2 elementi

//  ovverosia il numero massimo di parameter marker

exec sql

  allocate descriptor local :NomeDesc with max 2;



// costruzione istruzione SQL e impostazione elementi del descrittore

sqlstm = 'select EMPNO, LASTNAME, FIRSTNME from EMPLOYEE';

  // filtro codice impiegato

if WSEMPNO <> *blanks;

  if sqlstmW = *blanks;

    sqlstmW = ' where EMPNO >= ?';

  else;

    sqlstmW = sqlstmW + ' and EMPNO >= ?';

  endif;

  NumParm += 1;

   exec sql

    set descriptor :NomeDesc

       value :NumParm type = 1, length = 6, data = :WSEMPNO;

endif;

  // filtro ufficio

if WSDEPT <> *blanks;

  if sqlstmW = *blanks;

    sqlstmW = ' where WORKDEPT = ?';

  else;

    sqlstmW = sqlstmW + ' and WORKDEPT = ?';

  endif;

  NumParm += 1;

   exec sql

    set descriptor :NomeDesc

       value :NumParm type = 1, length = 3, data = :WSDEPT;

endif;

  // istruzione sql finale

sqlstm = sqlstm + sqlstmW + ' order by EMPNO';



  // numero totale di parameter marker ovverosia numero totale

  //  di elementi aggiunti al descrittore

exec sql

  set descriptor :NomeDesc count = :NumParm;



// preparazione

exec sql

  prepare pCurs1 from :sqlstm;



// apertura cursore utilizzando il descrittore SQL

//  per valorizzare i parameter markers

exec sql

  open Curs1 using sql descriptor :NomeDesc;



dou sqlcode = 100;

  exec sql

    fetch next from Curs1 into :dsOut;



  if sqlcode = 100;

    leave;

  endif;



  Msg = dsOut.Codice + ' ' + %trim(dsOut.Cognome) + ' ' + %trim(dsOut.Nome);

  dsply Msg;



enddo;



exec sql

  close Curs1;



exec sql

  deallocate descriptor local :NomeDesc;



*inlr = *on;

